
個人サポータ制度のお知らせ 
JGGUGでは, 2011年度より個人サポータ制度を始めました.

今までJGGUGの経済的な基盤はすべて法人会員の年会費に依存していました。 しかし、 個人会員も金

銭面からサポートしたいという声があがり、JGGUG運営基盤の裾野を拡げるためにも、個人サポータとい

う形で個人会員からの寄付を受け入れることにした次第です。

個人サポータとなっていただいた方には、 一年間にわたってJGGUGが発行する G* Magazine（年数回刊。

基本的に電子版として配布予定）に個人サポータとしてお名前を掲載します（掲載を希望しない旨お申し

出いただけば掲載しません）。

個人サポータとなるには、まず supporters@jggug.org にメイルで

•	お名前

•	予定金額

•	G* Magazineへのご芳名掲載の可否

をお知らせください。追って、運営委員より振込先の情報などを返信します。

皆様のサポートをお待ちしております。

日本 Grails/Groovy ユーザーグループ

代表　山田 正樹

supporters@jggug.org

山本　剛（株式会社ニューキャスト）
出版・印刷関連のシステム設計開発等に従事するテクニカル DTP アーキテクト。
日本 Grails/Groovy ユーザーグループ名古屋支部長。2006 年より Grails のドキュメント翻訳、
その後、Grails 公式の Acegi プラグインを開発。書籍『Grails 徹底入門』( 翔泳社発行 ) の 9、10、11 章を執筆。

緊 急 企 画 Grails 1.4.0 M1リリース

検証 ! Grails 1.4.0 の世界

今回は、「Grails 1.4.0 M1 リリース」緊急企画として、楽しみに
してた方には申し訳ないですが、前回の続き「Grails をコントロー
ルせよ！ Part 2」をお休みして、先日リリースされた、Grails 1.4.0 
M1 の追加機能・変更点など一部を紹介します。

Grails 1.4.x 系最初のリリース！
去年に発表されたロードマップの情報から、様々な先行情報が

見えてきた、Grails 次期バージョン 1.4 系の最初のマイルストー
ンがリリースされました。Grails 1.4.x 系では変更点が多く、内部
的にはかなり深いところまで改善・向上されています。今回の更
新内容を公式サイトのロードマップから内容を拝借しますと、

•	 �Groovy 1.8, Spring 3.1 M1, Hibernate 3.6,Servlet 3.0, Tomcat 7
への更新。

•	 エージェントベースリロード
•	 プラグイン使用統計トラッキング
•	 AST 変換を使用した GORM API と Java 統合
•	 HTML5 で強化したスカッフォルド UI
•	 静的リソースハンドリングの向上
•	 Mixin ベースの Unit テストサポート
•	 バイナリープラグイン
•	 Prototype から jQuery への変更
•	 基本データベースの H2 への変更
•	 GORM の機能強化

これだけあると、今回の記事だけでは全ての情報を掲載できま
せん。今回は、このマイルストーンで著者が気になった部分の一
部を紹介します。今後も、残りの情報・最新情報があればできる
限り記事にしていきたいと思っています。では、早速見ていきま
しょう。

またしても UIデザインの変更！
これで３回目となる、ベースデザインの変更です。最初の青っ

ぽいデザインから、次にリニューアルされた Spring カラーの UI、
そして時代に合わせてか今回は Spring カラーを採用しながらの
HTML5 化です。個人的な意見では、CSS の修正だけでもよかっ
たのでは？と思っていますが・・・。スクリーンショットを掲載
しておきます。

今回の変更では HTML5 を採用したデザイン内容になっていま
す。中身は Modernizr （http://www.modernizr.com/）が使われて
います。大幅に変わりすぎて、いや、違和感がありすぎて別物に
見えたりします。iPhone の Safari 用の調整もしてあるので、iOS
でも上手くはまっています。apple-touch-icon も定義してあるの
で、iOS でのアイコンもバッチリです。

個人的な意見は多々あると思いますが、別テンプレートを作っ
てしまおうかと。ってのはウソで、がんばって慣れようかと思い
ます。

[ 新デザイン ] [iOS 画面 ]

[iOS アイコン ]

緊急企画　検証 !  Grails 1.4.0 の世界

H2 になっただけでは無くて！データベース
コンソールもつかえる！

1.4.0 からデータベースが、HSQLDB に代わり H2 になりました。
そして、ただ H2 になっただけでは無く、H2 付属のデータベース
管理コンソールも使用できるようになりました。使用方法は単
純に、アプリケーションの URL に、/dbconsole と追加するだけ
です。例えば、http://localhost:8080/t140m1/ であれば、http://
localhost:8080/t140m1/dbconsole/ となります。管理コンソール
を参照すると、DB の指定、ドライバー、JDBC URL 等を設定する
ログイン画面が現れます。ここまで来たら、後は接続ボタンを押
すだけ！と行きたいところですが、状況によっては、JDBC URL
がデフォルトの状態になっている場合があるので設定します。

JDBC URL・ユーザ名・パスワード等の設定を行って接続する
と、SQL の実行を行ったりとデータベースの操作ができる管理コ
ンソールが現れます。これは便利！

デ ー タ ベ ー ス 管 理 コ ン ソ ー ル は、 基 本 的 に は 開 発
(development) モードでしか動作しません。場合によっては、本
運用環境でも管理コンソールを使用したいケースもあります。そ
んな時は Config.groovy の環境別設定 (environments ブロック )
の production 以下に、"grails.dbconsole.enabled = true" と、URL
の指定 "grails.dbconsole.urlRoot = '/admin/dbconsole'" を指定し

ます。本運用で使用する際はセキュリティを考慮して、必ず、
spring-security-core プラグインなどでアクセス制御しましょう。

environments {

    production {

        grails.serverURL = "http://www.changeme.com"

        grails.dbconsole.enabled = true

        grails.dbconsole.urlRoot = '/admin/dbconsole'

    }

    development {

        grails.serverURL = 

"http://localhost:8080/${appName}"

    }

    test {

        grails.serverURL = 

"http://localhost:8080/${appName}"

    }

}

[DB 管理コンソールログイン画面 ]

[ コンソール画面 ]

緊急企画　検証 !  Grails 1.4.0 の世界

Grails 1.4 からの Unit テスト
今回のリリースで、Unit テストが大幅改良されて、今まで可能

では無かった部分まで、Unitテストでテストを書くことができます。
例えば、インテグレーションテストが必要だったコントローラ

関連の、フィルタ−・URL マッピング・ファイルアップロード・
コンテントネゴシエーション・テンプレートビューレンダリング
や、ドメインクラスでは、クライテリアクエリーなども Unit テ
ストでサポートされています。

新 Unit テストの一部を試してみたので紹介します。詳しくは
1.4.x 系のドキュメントを参考にしてください。

■コントローラの Unit テスト
Grails 1.4.x 系の Unit テストでは、従来の GrailsUnitTestCase を

使用しません。下位互換用にクラスは存在していますが非推奨で
す。代わりに、アノテーション grails.test.mixin.TestFor にテスト
するコントローラクラスを指定します。このアノテーションを指
定する事によって、grails.test.mixin.web.ControllerUnitTestMixin
と関連する API が自動で設定されます。

import grails.test.mixin.*
import org.junit.*

@TestFor(HomeController)
class HomeControllerTests {

    void testSomething() {
        //ここにテストを書くよ。
    }
}

さらには、今まで、在って当然と思っていた静的スカッフォル
ドで生成されたコントローラの Unit テストが、grails generate-all 
コマンドで同時に生成されるようになりました。

生成されたコントローラの Unit テストは、完全な形ではなく、
必要な ( 後で変更があるであろう ) ドメインクラスやパラメータ
の部分は TODO: と記載されており、その部分は実装に応じて自
分で記述する必要があります。

例えば myapp.Book といったドメインクラスがあるとします。

package myapp

class Book {
    String title
    String author
    String isbn
    Long price
    static constraints = {
        title()
        author()
        isbn minSize:9
        price()
    }
}

静的スカッフォルドを生成します。

% grails generate-all myapp.Book

生成された Unit テストのコード BookControllerTests.groovy は
次のようになります。( 一部抜粋 )

生成されたコードの中に TODO: と書かれた部分があるので、
その辺りにドメインクラスの内容に沿ったコードを追記します。
( ①の部分 )

import org.junit.*

import grails.test.mixin.*

@TestFor(BookController)

@Mock(Book)

class BookControllerTests {

    @Test

    void testIndex() {

        controller.index()

        assert "/book/list" == response.redirectedUrl

    }

    ..... 中略 .....

    @Test

    void testSave() {

        controller.save()

        assert model.bookInstance != null

        assert view == '/book/create'

        // TODO: Populate valid properties

        // ①ここの部分に追記
        controller.save()

        assert response.redirectedUrl == '/book/show/1'

        assert controller.flash.message != null

        assert Book.count() == 1

    }

｝

サ ン プ ル コ ー ド 内 ① の 部 分 が 含 ま れ て い る テ ス ト は、
testSave() なので、フォームの内容を投げて保存するといった動
作になります。ここの部分に必要なのは、フォームに入るべきパ
ラメータ、アクションでの params になります。なので次のコー
ドように、サンプルコード内①以下に、params マップにフォー
ムに実際に入力する値をセットします。サンプルコード内②の 
controller.save() が実行されることによって、save アクションの
テストが行えます。

緊急企画　検証 !  Grails 1.4.0 の世界

@Test
void testSave() {
    controller.save()

    assert model.bookInstance != null
    assert view == '/book/create'

    // TODO: Populate valid properties
    // ①ここの部分に追記
    params.title='吾輩は猫'
    params.author='夏目漱石'
    params.isbn='0-00000000-000000'
    params.price=1000

    controller.save()//②

    assert response.redirectedUrl == '/book/show/1'
    assert controller.flash.message != null
    assert Book.count() == 1
}

他のアクション、例えば show、edit 等の予めドメインクラス
が必要になるコードの場合も似たような追記をします。

アクション show の例を紹介します。今度は、(a) の部分に通常
行うのと同じように、Book のインスタンスを生成して、それぞ
れのプロパティを設定します。そして、book.save() を実行する
ことで、Unit テストで内部的に保存され Book の id が取得できま
す。params.id に、Book の id を設定して、controller.show() を実
行することで、show アクションのテストが完成です。

@Test
void testShow() {
    controller.show()

    assert flash.message != null
    assert response.redirectedUrl == '/book/list'

    // TODO: populate domain properties (a)
    def book = new Book()
    book.title='吾輩は猫'
    book.author='夏目漱石'
    book.isbn='0-00000000-000000'
    book.price=1000

    assert book.save() != null

    params.id = book.id

    def model = controller.show()

    assert model.bookInstance == book
}

ここで気になるのは、Book クラスの存在です。先ほど、生成
されたコードのアノテーション部分をみると @Mock(Book) と指
定してあります。この指定で Book クラスのモックが定義されて

います。それによって、この同一 Unit テスト内では、Book を扱
うことができるのです。

import org.junit.*

import grails.test.mixin.*

@TestFor(BookController)

@Mock(Book)

class BookControllerTests {

..... 省略 .....

@Mock の指定は、複数のドメインクラスを指定する事も可能
です。その場合は @Mock([Book, Author]) と、配列で定義するだ
けです。

@Mock を 指 定 す る 事 で、 ド メ イ ン ク ラ ス を デ ー
タ ベ ー ス に 繫 ぐ こ と な く GORM の 機 能 を テ ス ト で
き る、DomainClassUnitTestMixin が 設 定 さ れ ま す。
DomainClassUnitTestMixin を 使 用 す る と メ モ リ 上 の
ConcurrentHashMap 内で GORM のデータが構築され動作します。

@TestMixin で、 直 接 DomainClassUnitTestMixin を 定 義 す る
ことも可能です。この場合は、ドメインクラスのモックは、
mockDomain( ドメインクラス名 ) で直接生成します。

@TestFor(BookController)

@TestMixin(DomainClassUnitTestMixin)

class BookControllerTests {

    void testSave() {

         mockDomain(Author)

         mockDomain(Book)    

    }

}

■ドメインクラスの Unit テスト
ここまででわかるように、コントローラ Unit テストでドメイ

ンクラスの Unit テストも簡単に行えるのがわかります。ただ、
やはり制約などは、ドメインクラス用のテストを書きたいと思い
ます。従来通りドメインクラスを create-domain-class コマンド
で生成すると、ドメインクラスの Unit テストコードも生成され
ます。但し内容は以前と同じ何も記述されていない状態です。

先ほどのドメインクラス Book の Unit テストの例を紹介しま
す。※誌面の関係上今回は多く説明しません。今後の参考になれ
ばと思います。

緊急企画　検証 !  Grails 1.4.0 の世界

import grails.test.mixin.*

import org.junit.*

@TestFor(Book)

class BookTests {

    void testSomething() {

        def book = new Book()

        book.title='吾輩は猫'

        book.author='夏目漱石'

        book.isbn='123456789'

        book.price=1000

        assert book.save() != null

       

        assert book.id != null

        assert book.title=='吾輩は猫'

    }

   

    void testConstraints(){

        mockForConstraintsTests Book

       

        def book = new Book()

        assert !book.validate()

        assert "nullable" == book.errors["title"]

        assert "nullable" == book.errors["author"]

       

        book.title='吾輩は猫'

        book.author='夏目漱石'

       

        book.isbn='12345678'

        assert !book.validate()

        assert "minSize" == book.errors["isbn"]

       

        assert !book.validate()

       

        book.isbn='123456789'

        book.price=1000

        assert book.validate()

    }

}

最後にテストレポートもデザインが変更になったので、スク
リーンショット載せておきます。

[ テストレポート ]

今回は、UI・DB・Unit テストを紹介しました。まだ正式版の
リリースは時期が 2011 Q3 としかわかっていません。これから
M2,M3? と 1.4 系のマイルストーンリリースラッシュになるのは
確かです。速度・機能と共に向上した Grails 1.4 系、とても楽し
みですね。

次号からも、時間がある限り、複数の著者で順番に「検証！
Grails 1.4.0 の世界。」シリーズを掲載していきたいと思います。

緊急企画　検証 !  Grails 1.4.0 の世界

G* なみなさま、こんにちは。今回からスレッド編として 2 回
に分けて Griffon アプリケーションでのスレッドの扱い方を紹
介していきます。今回は Swing アプリケーションでスレッドを
扱うための基本となる Swing のスレッドポリシーと Groovy の
SwingBuilder でスレッドを扱う方法を紹介し、次回は Griffon で
のスレッドの使い方をサンプルアプリケーションを作りながら紹
介したいと思います。

Web アプリケーションの開発がメインで Swing アプリケー
ションにあまり馴染みのない方は、これまでにスレッドを扱う機
会が少なかったかと思います。Web アプリケーションでは独自
のスレッドを生成して処理するということはあまり必要とされま
せん。しかし、Swing アプリケーションではスレッドの扱いを理
解しておくことが大変重要になります。

Swing のスレッドポリシー
Swing アプリケーションで発生するイベントはすべて、イベン

トディスパッチスレッドと呼ばれる 1 つのスレッドで処理されま
す。

また、基本的に Swing コンポーネントはスレッドセーフでは
ないため、Swing コンポーネントに対する操作はすべてイベント
ディスパッチスレッド上で実行する必要があります。「基本的に」
と言いましたが、例外もあります。例えば、テキストエリアにテ
キストを追加する javax.swing.JTextArea クラスの append(String)
メソッドはスレッドセーフになっています。どのメソッドがス
レッドセーフかどうかは Javadoc に記述されています。例えば
javax.swing.JTextArea の append(String) メソッドの Javadoc には
以下の注意書きが記述されています。

このメソッドはスレッドに対して安全ですが、ほとんどの Swing メ
ソッドは違います。

スレッドセーフなメソッドをイベントディスパッチスレッド内
で呼び出しても問題ありませんが、スレッドセーフではないメ
ソッドをイベントディスパッチスレッド外で呼び出した場合は、
予期しない動作をすることがあるでしょう。スレッドセーフに関
する問題は、再現性がないため慎重に実装する必要があります。

イベントディスパッチスレッド上で処理を行うにはいくつか
方法があります。まず、各 Swing コンポーネントに登録したイ
ベントリスナの各アクションイベントを受け取るメソッド (java.
awt.event.ActionListener クラスの actionPerformed() メソッドな
ど ) はイベントディスパッチスレッド上で実行されます。イベン
トディスパッチスレッド外から Swing コンポーネントに対する
操作を行う場合は javax.swing.SwingUtilities に用意されている

static メソッドを利用してイベントディスパッチスレッド上で処
理を行うようにスケジュールします。

SwingUtilities
SwingUtilities に用意されているイベントディスパッチスレッ

ド上で処理をするためのメソッドには invokeLater(Runnable)、
invokeAndWait(Runnable) があります。それぞれイベントディス
パッチスレッド上で実行する処理を java.lang.Runnable インター
フェースに実装したクラスのインスタンスを引数に取ります。
invokeLater メソッドはイベントディスパッチスレッドにスケ
ジュールします。スケジュールされた処理は、スケジュールされ
た順番に実行されます。invokeLater メソッドはスケジュールし
たあとで呼び出し元に処理が戻りますが、invokeAndWait メソッ
ドはスケジュールした処理が実行されるのを待ってから呼び出し
元に戻ります。次のコードでそれぞれのメソッドの動作を確認で
きます。

▼invokeLater_sample.groovy

import javax.swing.SwingUtilities

import groovy.swing.SwingBuilder

println "1 - ${Thread.currentThread().name}"

new SwingBuilder().edt() {

    frame(show:true, pack:true) {

        button 'Click', actionPerformed: {

            doOutside {

                SwingUtilities.invokeLater {

                    sleep 1000

                    println "2 - ${Thread.

currentThread().name}"

                }

                println "3 - ${Thread.

currentThread().name}"

            }

        }

    }

}

Griffon 不定期便
 ～第 3回 スレッド編 その１～

02
ser ies 奥  清隆 （おく きよたか）

仕事でもときどき Groovy と戯れるプログラマ。
日本 Grails/Groovy ユーザーグループ関西支部長。
著書：『Seasar2 による Web アプリケーションスーパーサンプル』

Griffon 不定期便 〜第 3 回 スレッド編 その１〜

▼invokeAndWait_sample.groovy

import javax.swing.SwingUtilities

import groovy.swing.SwingBuilder

println "1 - ${Thread.currentThread().name}"

new SwingBuilder().edt() {

    frame(show:true, pack:true) {

        button 'Click', actionPerformed: {

            doOutside {

                SwingUtilities.invokeAndWait {

                    sleep 1000

                    println "2 - ${Thread.

currentThread().name}"

                }

                println "3 - ${Thread.

currentThread().name}"

            }

        }

    }

}

両方とも、同じウィンドウが表示されますが、ボタンをクリッ
クしたときの挙動が異なります。それぞれ doOutside() を使用し
てイベントディスパッチスレッド外から SwingUtilities クラスの
メソッドを呼び出すようにしています。（doOutside() メソッドに
ついては後で説明します。）

最初のサンプルコードを実行し、ボタンをクリックすると標準
出力に次のように表示されます。

1 - main

3 - Thread-4

2 - AWT-EventQueue-0

invokeLater() メソッドに渡した処理を待たずに 3 番目の println
が実行されているのがわかります。invokeAndWait を使用するサ
ンプルコードの実行結果は次のようになります。

1 - main

2 - AWT-EventQueue-0

3 - Thread-4

こちらは、invokeAndWait() メソッドに渡した処理が実行され
て終わるのを待ってから 3 番目の println が実行されているのがわ
かります。イベントディスパッチスレッド外から Swing コンポー
ネントを操作する場合は状況に合わせてこの 2 つのメソッドを使
用することでスレッドセーフなアプリケーションになります。

すべての処理をイベントディスパッチスレッド上で実行すれば
スレッドセーフに関する問題はなくなりますが、時間のかかる処
理をイベントディスパッチスレッド上で実行するとアプリケー
ションがフリーズしたような状態になります。

次のようなコードを実行してみましょう。

import groovy.swing.SwingBuilder

new SwingBuilder().edt {

    frame(show:true, pack:true) {

        gridLayout(cols:1, rows:2)

        label id:'label'

        button '更新', id:'button', 

actionPerformed: {

            sleep 3000

            label.text = new Date().

format('HH時mm分ss秒')

            println 'Clicked!'

        }

    }

}

更 新 ボ タ ン を ク リ ッ ク す る と、3 秒 後 に 時 刻 が 表 示 さ
れ、 標 準 出 力 に「Clicked!」 と 出 力 さ れ ま す。javax.swing.
JLabel#setText(String) メソッドは、スレッドセーフではありませ
んが、ボタンの actionPerformed に設定したクロージャはイベン
トディスパッチスレッド内で実行されるため、このアプリケー
ションはスレッドセーフです。しかし、更新ボタンをクリックし
たときにボタンが押下状態のままになり、アプリケーションがフ
リーズしているように見えます。

押下状態のままボタンをクリックしても動いていないように見
えますが、実はクリックしたときに発生するイベントがイベント
ディスパッチ上にスケジュールされています。しばらく押下状態
が続いたのち、クリックした回数だけ標準出力に「Clicked!」と
表示されます。もしこれが何かデータを登録する処理だった場合、
ユーザがクリックした回数だけデータが登録されてしまいます。
このようにアプリケーションが止まっているような状態になるの
を回避するためには、イベントディスパッチスレッドから新しく
スレッドを生成して処理する必要があります。この方法について
は後ほど紹介いたします。

Swing のスレッドポリシーについては Javadoc に記述されてい
ますので、一度読んでおくといいでしょう。

http://java.sun.com/javase/ja/6/docs/ja/api/javax/swing/
package-summary.html#threading

Griffon 不定期便 〜第 3 回 スレッド編 その１〜

SwingBuilder でスレッドを扱う
Swing でのスレッドの扱い方を紹介しましたが、Groovy の

SwingBuilder ではスレッドの扱いが簡単でさらに賢く使えるよう
になっています。SwingBuilder ではスレッドを扱うメソッドと
して、edt()、doOutside()、doLater() の 3 つが用意されています。
それぞれのメソッドは SwingUtilities と以下のように対応します。

•	 edt()
	� SwingUtilities#invokeAndWait() に相当。イベントディ

スパッチスレッド上で呼び出された場合は引数のク
ロージャをそのまま実行。

•	 doLater()
	 SwingUtilities#invokeLater() に相当。
•	 doOutside()
	� イベントディスパッチスレッド上で呼び出された場合

は、新しいスレッドを生成して、引数のクロージャを
イベントディスパッチスレッド外で実行。イベントディ
スパッチスレッド外で呼び出された場合はクロージャ
をそのまま実行。

これらのメソッドを使わずに、SwingUtilities を使用するこ
ともできますが、SwingBuilder のメソッドを使用することで
処理がどのスレッドで実行しているかが分かりやすくなりま
す。 ま た SwingUtilities#invokeAndWait() メ ソ ッ ド は イ ベ ン ト
ディスパッチスレッド内から呼び出すとエラーが発生しますが、
SwingBuilder の edt() メソッドは呼び出されたスレッドがイベン
トディスパッチスレッドかどうかを判断してくれるので使い勝手
もよくなります。

SwingBuilder を使用して先ほどのサンプルコードを正しく実装
し直してみましょう。

import groovy.swing.SwingBuilder

new SwingBuilder().edt {
    frame(show:true, pack:true) {
        gridLayout(cols:1, rows:2)
        label id:'label'
        button '更新', id:'button', 

actionPerformed: {
            // ①
            button.enabled = false
            label.text = ''
            // ②
            doOutside {
                3.times {
                    // ③
                    edt { label.text += '.' }
                    sleep 1000
                }
                // ④
                doLater {
                    label.text = new Date().

format('HH時mm分ss秒')
                    button.enabled = true
                    println 'Clicked!'
                }
            }
        }
    }
}

ボタンをクリックしたときのイベントはイベントディスパッチ
スレッド上で実行されます。処理が終わるまでボタンをクリック
出来ない様に javax.swing.JButton#setEnabled(boolean) メソッド
を使用してボタンを無効化しています（①）。

次に時間のかかる処理をバックグラウンドスレッドで実行する
ように doOutside を使用しています（②）。doOutside に渡すク
ロージャはイベントディスパッチスレッドとは別の新しいスレッ
ドで実行されます。3 秒待つために、1 秒間スレッドを止める処
理を 3 回ループしています。このループの中でラベルのテキスト
に「.」を追加しています（※）。これはユーザに処理中であるこ
とを知らせる簡易的なプログレスバーと考えてください。このと
き、処理はバックグラウンドスレッド内で実行されているので、
イベントディスパッチスレッド上でラベルのテキストを変更する
必要があります。ラベルのテキストを変更したあとでスレッドを
止めるには edt() メソッドを使用します（③）。doLater() を使用し
てもイベントディスパッチスレッド上で実行されますが、処理を
スケジュールするだけですので後続のスレッド停止処理と並行し
て実行されることになります。

3 秒スレッドを停止したら、ラベルに時刻を表示して、ボタン
を有効化させます（④）。ここでは doLater を使用して処理をス
ケジュールさせます。バックグラウンドスレッドはこれで終了し
ますが、後でイベントディスパッチスレッド上でラベルに時刻が
表示され、ボタンがクリック出来る状態になります。

※ボタンを無効化させるだけでもアプリケーションとして
は問題ありませんが、それだけでは待たされる方として
はアプリケーションが動いているのかどうかわかりませ
ん。今回は説明をシンプルにするためにラベルのテキスト
を変更していますが、実際にアプリケーションを作成す
る 場 合 は javax.swing.JProgressBar な ど を 使 用 し て 処 理 の
進捗状況を知らせるのが良いでしょう。SwingBuilder では
progressBar() メソッドを使用して javax.swing.JProgressBar
コンポーネントを利用できます。

Griffon 不定期便 〜第 3 回 スレッド編 その１〜

まとめ
イメージとしては、仕事中の自分がイベントディスパッチス

レッドと考えてください。何か作業が発生（イベントが発生）し、
その作業を他の誰かに振るのが doOutside() です。作業を他の人
に振ったので別の作業に取り掛かれます。作業を振られた人が作
業の途中であなたに確認を取らなければいけなくなり、「確認お
願いします」と言い、あなたは「そこに置いといて下さい、手が
開いたときにやっときます。」というのが edt() です。確認をお願
いした方は確認が終わるまで作業を終えることができませんし、
他のこともできません（ひどい話ですが例えばの話です）。

また別の作業があって、その作業は誰にも振らず自分でやると
します。自分でやりますが別の作業があって今やれないときは「あ
とでやる」ことにします。これが doLater() です。

あまり関係ないですが、私は「あとでやる」というのが好きです。
あとでやるというか「明日やる」ことにするのが好きです。そし
てギリギリになってやる羽目になり泣きながらやります。この記
事を書いている今も少し泣いています。

少しややこしくなりましたが、Swing のスレッドポリシー
を理解することでアプリケーションの出来が変わってきます。
SwingBuilder の edt()、doLater()、doOutside() の使い方をマスター
すれば、より簡単にスレッドを扱うことが可能になります。今
回は Swing のスレッドポリシーと Groovy で使う方法の紹介で終
わってしまいましたが、次回はいよいよ Griffon でスレッドを扱
う方法を紹介いたします。それではまた次回。

Griffon 不定期便 〜第 3 回 スレッド編 その１〜

はじめに
前回は CodeNarc の基本操作を解説した。
今回は開発の現場でより実践的に利用できるようにするために

方法として次の項目を解説する。

•	 IntelliJ IDEA との統合
•	 Grails プラグインの利用

最終回の第 3 回は解析ルールのカスタマイズ方法を解説する。

本稿は次の環境を前提で話を進める。
•	 Groovy1.7 がインストールされていること
•	 JDK1.6 がインストールされていること

IntelliJ IDEAとの統合

■IntelliJ IDEA のインストール
IntelliJ IDEA は、チェコの JetBrains 社により開発されている統

合開発環境である。Java の開発は有償版でないと実施できない
が Groovy の開発は Community 版で可能である。機能の詳細等は
Web 等の記事を参考にして頂きたい。

Community 版 は、http://www.jetbrains.com/idea/download/
index.html からダウンロード可能である。インストールはインス
トーラーの指示に従って簡単にインストールすることができるの
で各自でインストールして頂きたい。

■CodeNarc プラグインの組込み
IntelliJ IDEA を起動し、_File_ → Settings の順番にクリックし左

側ペインで Plugins を選択する。右側ペインで Available タブをク
リックし CodeNarc を選択し右クリックし Download and Install
をクリックする。ダウンロード及びインストールが実行された後
に OK をクリックすると再起動をうながすポップアップがあがる
ので再起動する。

再 起 動 後、File → Settings を ク リ ッ ク し 右 側 ペ イ ン で
Inspections をクリックすると右側ペインに CodeNarc が表示され
ているのでチェックをいれる ( 図 1)。

[ 図 1]

■解析の実行
Inspections は有効にすると即座にコードに反映される。問題

を検知した行はエディタの右側に黄色で表示され、該当行はうす
いオレンジ色で反転される。反転されている場所にカーソルを合
わせるとライトのマークが左に表示されクリックすると適用され
ているルールや変更適用などの項目などが表示される ( 図 2)。

[ 図 2]

CodeNarcを利用してGROOVYのコード品質を上げる
～第 2回 開発ツールと連携～

03
ser ies 荒井健太郎

某 IT ベンダーのセキュリティコンサルタント、ソフトウエアをよりセキュアにするために日々精進しています。
テスト系スクリプトは、G* を利用して書いています。

CodeNarc を利用して GROOVY のコード品質を上げる〜第 2 回　開発ツールと連携〜

Grails プラグインの利用
Grails の CodeNarc プラグインを利用すると、Grails で作成され

た Web アプリケーションに対して容易に CodeNarc の解析を実行
することができる。

本稿は Grails を利用して作成したアプリケーションが既にある
前提で話を進める。Grails の詳細については、G*Magazine 準備
号の「Grails - JavaEE 開発をライトにするフルスタックフレーム
ワーク」等の記事を参照して頂きたい。

■Grails プラグインの導入
Grails プロジェクトディレクトリに移動し次のコマンドを実行

する。

grails install-plugin codenarc

Grails のスクリプトが実行されプラグインが導入される。

■Grails プラグインの実行
Grails プロジェクトディレクトリに移動し次のコマンドを実行

する。

grails install-plugin codenarc

CodeNarc の 解 析 が Grails プ ロ ジ ェ ク ト に 対 し て 実 行 さ れ、
CodeNarcReport.html がプロジェクトディレクトリに作成され
る。

■Grails プラグインの設定
●標準の設定

Grails プ ラ グ イ ン は 標 準 の 設 定 で Grails プ ロ ジ ェ ク ト の
次 の デ ィ レ ク ト リ に 対 し て 解 析 を 実 行 し、HTML 形 式 の
CodeNarcReport.html レポートを生成する。

•	 src/groovy
•	 grails-app/controllers
•	 grails-app/domain
•	 grails-app/services
•	 grails-app/taglib
•	 grails-app/utils
•	 test/unit
•	 test/integration
また、解析に利用されるルールは、rulesets/basic.xml,rulesets/

exceptions.xml, rulesets/imports.xml,rulesets/grails.xml, rulesets/
unused.xml である。

設 定 は、_grails-app/conf/config.groovy_ に _codenarc.XXX_
を追加し変更できる。代表的なものを次節以降で紹介する。

●レポートのカスタマイズ
出力レポートのカスタマイズは次の書式で変更する。

codenarc.reports = {

　　　それぞれのレポートに対して以下の設定を実施する:

        REPORT-NAME(レポート形式) {

            プロパティ名 = プロパティ値
            プロパティ名 = プロパティ値
        }

}

レ ポ ー ト 形 式 は、_html_, _xml_, _text_, CodeNarc の
ReportWriter を継承したクラス名を指定できる。

次は、XML 形式でファイル名が CodeNarc-Report.xml、タイ
トルが Sample Report のレポートと HTML 形式でファイル名が
CodeNarc-Report.html、タイトルが Sample Report のレポートを
出力する例である。プロパティを編集し Grails プラグインを実行
すると、設定した値でレポートが作成されることが確認できる。

codenarc.reports = {

    MyXmlReport('xml') {                    

// The report name "MyXmlReport" is user-defined; 

Report type is 'xml'

        outputFile = 'CodeNarc-Report.xml'  

// Set the 'outputFile' property of the (XML) 

Report

        title = 'Sample Report'             

// Set the 'title' property of the (XML) Report

    }

    MyHtmlReport('html') {                  

// Report type is 'html'

        outputFile = 'CodeNarc-Report.html'

        title = 'Sample Report'

    }

}

●ルールのカスタマイズ
利用するルールのカスタマイズは次の書式で変更する。

codenarc.ruleSetFiles="file:利用するルールファイル"

次は、カスタムルール ( 次回に解説予定 )MyRuleSet.groovy を
利用する例である。

codenarc.ruleSetFiles=

"file:grails-app/conf/MyRuleSet.groovy"

●解析対象ファイルのカスタマイズ
•	 codenarc.processSrcGroovy - src/groovy を 解 析 対 象 と す る

か。 ( 標準は true)
•	 codenarc.processControllers - grails-app/controllers を解析対

象とするかを指定。 ( 標準は true)
•	 codenarc.processDomain - grails-app/domain を解析対象と

するかを指定。 ( 標準は true)

CodeNarc を利用して GROOVY のコード品質を上げる〜第 2 回　開発ツールと連携〜

•	 codenarc.processServices - grails-app/services を解析対象と
するかを指定。 ( 標準は true)

•	 codenarc.processTaglib - grails-app/taglib を解析対象とする
かを指定。 ( 標準は true)

•	 codenarc.processTestUnit - test/unit を解析対象とするかを
指定 ( 標準は true)

•	 codenarc.processTestIntegration - test/integration を 解 析 対
象とするかを指定 ( 標準は true)

•	 codenarc.processViews - grails-app/views 以下の GSP を解析
対象とするかを指定 ( 標準は false)

•	 codenarc.extraIncludeDirs - 上記以外のソースを解析対象と
する場合に指定

次は、'grails-app/jobs/*.groovy' を解析対象とする場合の例で
ある。

codenarc.extraIncludeDirs=['grails-app/jobs']

■まとめ
今回は、CodeNarc を IntelliJ IDEA と統合する方法及び Grails プ

ラグインを利用して静的解析する方法を解説した。これによって、
CodeNarc がより現場で利用されるようになればと考えている。

今回紹介した以外にもさまざまなフレームワーク、ツールと連
携することが可能である。詳細は、http://codenarc.sourceforge.
net/codenarc-other-tools-frameworks.html を参照して頂きたい。
次回はプロジェクト固有のルールを作成する方法を解説し最終回
とする予定である。

CodeNarc を利用して GROOVY のコード品質を上げる〜第 2 回　開発ツールと連携〜

第１章　�もかは『Groovy イン・アクション』
と出会った

■新人女子 Javaプログラマ
4 月、それは新社会人にとって新たな幕開けの季節だ。もかに

とってもそれはそうだった。七海 萌香 ( ななみ もか )、はソフト
ハウスとしては中堅どころのジェイガ株式会社へとプログラマ志
望で入社した新入社員の一人だった。

新人教育でビジネスマナーやら簡単な手続きを済ませた後
は、早速プログラミング講習が待っていた。言語は Java だった。
もかはこれまで簡単なホームページを HTML や CSS、それから
Web1.0 時代の JavaScript で書いたことがあるくらいだったので、
これが本格的なプログラミング言語との出会いだった。

今年の新人は男性が８名、女性が２名であった。もう一人の新
人女子社員は一ノ瀬 小夏 ( いちのせ こなつ ) という名前の女の子
だった。こなつは情報系専門学校を卒業しており、当初はゲーム
開発に憧れて情報系に進んだらしいのだが、現実の厳しさなどを
目の当たりにし、それから色々あってジェイガに入ることに決め
たらしい。どうあれ、新人女子社員は二人だけということもあり、
もかとこなつはすぐに打ち解けることができた。

■ある日の定時後
　もか「あー今日も終わった終わった〜！お疲れちゃ〜ん！」
こなつ「はい...お、お疲れ様です」
　もか「へへーん、頑張ったわ私！　さてっと...ねぇねぇ、こ

なつぅ！ちょっといい？」

私と違ってこなつはちっちゃくて目もクリっとしててかわいい
んだよね、ついつい誘いたくなっちゃう。

こなつ	「え、あ、はい...何ですか？」
　もか	「今日ってさ、このあと時間ある？ どこかにパーッと遊

びに行きたいなぁって思ってるんだけど」

こなつ	「え、あ、はい...でも私ちょっと寄りたいところが...」

こなつは困った表情をしている。

　もか「あ〜そうなんだ...。ふーん、じゃあ仕方ないかぁ」
こなつ	「私、Javaの講義について行けてなくて...会社からも

らった参考資料だけじゃだめだなーって思うんです。だ

からちょっと本屋さんに寄って、Javaの本を買いたい

なって思ってるんです！」

　もか	「なるほどね、こなつは勉強熱心だよね、感心しちゃうよ」
こなつ「いえいえ、そんなことないです...私、出来が悪いし...み

んなについていくのに必死なんです」

それを聞いていたもかは目を閉じてしばし無言になった。しば
らく何か考えている様子だったが、ゆっくりと目を開いてからこ
う言った。

　もか「私も一緒について行っていい？」
こなつ「え、あ、はい...いいですよ」

きょとんとしているこなつを尻目に、もかは嬉しそうににんま
りと笑顔を見せた。

こなつは勉強熱心なのだが、どうやらプログラミングには自信
がないらしく苦労しているようである。会社でも Java の講義に
追いついていくのがやっとといったところのようである。だから
少しでも追いつこうとして、参考書を購入しようというわけだ。

　もか「本屋さんって近くにあるの？私、どうも本って苦手で
さぁ...一人では本屋さんなんてあまり行かないから」

こなつ「え、そ...そうなんですか？ごめんなさい、私誘ちゃっ
たみたいで...」

　もか「え？ううん、今日は私がついて行きたいって言い出した
んだし、こなつは全然悪くないよ！むしろ謝らなきゃい

けないのは私の方だよね」

こなつ「ううん、一緒に来てくれるの嬉しいですよ、一緒に選ん
でもらえるとすごく助かります」

こなつはえへっと笑顔を見せた。

 もし新人女子 Java プログラマが
『Groovy イン・アクション』を読んだら ～第１章～

吉田　健太郎
仕事では Java、趣味でライフワークの如く Groovy と戯れるプログラマ。
日本 Grails/Groovy ユーザーグループ運営委員参画予定。
ブログ：『No Programming, No Life』http://d.hatena.ne.jp/fumokmm/

もし新人女子 Java プログラマが『Groovy イン・アクション』を読んだら 〜第１章〜

■書店にて
　もか「さーて、着いたわね。こなつはもうどれを買うか決めて

るの？」

こなつ「え、えーと...か、簡単そうなのがいいなぁと思って...講
義の内容がちょっと難しくて。ほら、このあいだ、

classの話になったあたりから急に難しくなったでしょ

う？だから分かりやすく丁寧に書かれている参考書が欲

しいなって思ってるんです」

それを聞いたもかは、ある本を指差しながらこう言った。

　もか「ふーん、なるほどね。じゃあさ、これなんてどう？『た
のしいJava』っていうタイトルからしていかにもって

感じじゃない？」

もかはその本を取って渡してあげた。

こなつ「わぁ〜いいかもです。ちょっと見て見ますね」

こなつがその本を見ている間、もかは Java の本が並ぶ棚の他
の本を眺めて回っていた。

もかは Java こそ今回の研修で初めてやったとはいえ、プログ
ラミングというものにはずっと興味を抱いていたし、JavaScript
で簡単なアルゴリズムだったり、クラスだったりの概念は理解し
ているつもりである。今日まで行われた講義も分からなくなると
ころは全然なかった。

いっぱい本あるなぁ ... と、そんなことを思いながら何気なく本
を眺めていると、もかの視界にとある本が飛び込んできた。その
本のタイトルは、

『Groovyイン・アクション』
んっ？ぐるーびー？何だろうこれ。なんで Java のコーナーに？
他の本は Java という名前がつく本ばかりであるのに。気になっ

たのでちょっと手に取ってみた。ずっしりと重みを感じる。それ
もそのはず、600 ページ近くもあってかなり分厚い。表紙にはな
にやら日本人形のような踊り子さんが描かれている。帯にはこう
あった。

ふーん ... ？なんだかすごそうね。Python とか Ruby とかは聞い
た事はあるけど、Groovy ってどんな言語なんだろう？すもーる
とーくっていうのは知らないけど ...。ちょっと興味がわいたので

パラパラとページをめくってみた。

もかはふと P63 のコードに目が止まった。そのページのタイト
ルは「3.4.3 Java から Groovy へ」となっていた。

System.out.println("Hello Groovy!");

うんうん、これは Java のコードよね、一番初歩的はハローワー
ルドだわ。そしてすぐ下はこう続いていた。

System.out.println('Hello Groovy!');

ダブルクォーテーションがシングルクォーテーションになった
のね、JavaScript みたいなものかしら。

もかはそこまで気に留めず、例をさらに読み進めた。

print('Hello Groovy!');

すごいすっきりしてきたわ。Java だと System.out.println って
ドットでつないで長く書いていかないといダメけど、Groovy だ
と System.out が要らないんだ。

print 'Hello Groovy!'

わぁ、括弧までなくなっちゃった。すごくスッキリした感じ。
もかはさらに読み進めた。

greeting = 'Hello Groovy!'

assert greeting.startsWith('Hello')

assert greeting.getAt(0) == 'H'

assert greeting[0]       == 'H'

assert greeting.indexOf('Groovy') >= 0

assert greeting.contains('Groovy')

assert greeting[6..11] == 'Groovy'

assert 'Hi' + greeting - 'Hello' == 'Hi Groovy!'

assert greeting.count('o') == 3

assert 'x'.padLeft(3) == '  x'

assert 'x'.padRight(3,'_') == 'x__'

assert 'x'.center(3) == ' x '

assert 'x' * 3 == 'xxx'

え？ ほ ん と ? こ ん な 記 述 で い い の？ assert っ て あ れ？ こ な
いだの講義でやった JUnit ってやつにも出て来たやつだよね、
assertEquals とかってやつ。ってことはこれはつまり、テストコー
ドなんだわ！（後にもかは P28, P29 ページの assert の解説を読ん
で assertEquals よりも Groovy の assert がもっと軽量に使えて便利
だということを知ることになる）

え？うそうそ！？文字列に対してマイナスが使えるの？？あ
と [6..11] で切り出したりできるの？すごい！それから、最後の

●Groovy とは
Java プラットフォームのためのアジャイルで動的な言語です。
Python や Ruby、Smalltalk などの言語から多くの機能を取り
入れ、Java 開発者が Java 風の構文を使ってこれらを利用でき
るようになっています。スクリプト言語として言及されるこ
とも多いのですが、その枠にとどまるものではありません。
Java 実行環境上で動作するため、Java バイトコードにコンパ
イルすることも可能であるなど、非常に強い「Java との親和性」
を備えています。

もし新人女子 Java プログラマが『Groovy イン・アクション』を読んだら 〜第１章〜

padLeft, padRight, center っていうので文字を空白埋めしたり、
アンダーバーで埋めたりできちゃうんだ！それからそれから、文
字のかけ算も？

もかはすごく興奮しだしている自分を感じていたが、頭の中で、
これは Java じゃないんだから今の私とは関係ないないんだ！と
言い聞かせるようにしていた。

そうこうしているうちに、こなつがニコニコしながらもかの側
に寄って来た。手には先ほどの『たのしい Java』を持っている。

こなつ「もかちゃん！私この本にするね」
　もか「そうだね、その本ならJavaに限らずプログラミングの

基本的なところも勉強できるしいい本だと思うわ」

こなつ「私、この本読んで頑張る！みんなについて行けるように」
　もか「うんうん、頑張って勉強して一緒にかっこいいプログラ

マになろうねっ！」

こなつ「ありがとう、もかちゃん。あれ？ところでもかちゃんも
何か本を買うの？」

こなつはもかが持っている『Groovy イン・アクション』に目
がいった。

　もか「あー、えっと、なんかこれね、Javaとは違うプログラ
ミング言語の本なんだけど、ちょっと面白そうなのよ。

まだちょっと立ち読みしてただけだから、あまりわから

ないんだけどJavaとの親和性が高いんだって」

こなつ「えーー？すごいね、やっぱりもかちゃんは...私なんて
Javaだけで精一杯なのにもかちゃんは違うプログラミ

ング言語も勉強してるの？私どんどん置いて行かれちゃ

う...」

　もか「ちがうちがう、私もJavaは今回の講義で始めたばっ
かりだし、良くわかってないところが多いし...まずは

Javaをマスターしなきゃ！」

そう言いながら、もかは『Groovy イン・アクション』を棚に
戻した。

■自室にて
その日の夜。自室のベッドに入って寝る前にもかは今日本屋さ

んで見た『Groovy イン・アクション』のことを思い出していた。
なんとなく興奮が止まらない。プロブラミングは好きだけど、こ
んなにドキドキしたことって今までなかったな。もしかしたらこ
れって運命の出会い？とか言っちゃって。なにそれ ... 素敵な男の
子じゃなくてプログラミング言語との出会いなんて。もかはそ
んなことを考えながら苦笑する。もう ... こんなこと考えてるから
私ってモテないのかなぁ、女子力っていうの？全くない！こなつ
がうらやましいなぁ ... ちっちゃくて、かわいくて。ああいう子は
絶対にモテるんだよなぁ ...。

とりとめのないことを考えながら、いつのまにかもかは眠りに
ついていた。

■Java の演習
次の日、出社したもかたちは Java の講義を受けているところ

だ。今日も恒例の課題が出題されるようだ。

「さーて、みんなそろそろ Java に慣れてきたところだろう、次
はこんな問題だ。周りと相談しないで自分だけで解くように」

正の整数nが与えられたときに、高さnのピラミッドを出力
するプログラムを作ってください。
n=4の時の出力は下のようになります。

   *
  ***
 *****
*******

もかはピンときた。なるほどね、これは単純な文字列操作の例
ね。よーし、作るぞぉ！もかは意気込んでこの課題に取り組んだ。
数十分後、Java のコードが出来上がった。

public class Pyramid {
    /**
     * ピラミッドを出力するプログラム♪
     * @author 七海 萌花
     * @param args 何段にするか
     */
    public static void main(String[] args) {
        int dan = Integer.parseInt(args[0]);
        for (int i = 0; i < dan; i++) {
            String hidari = "";
            String migi = "";
            for (int j = 0; j < dan - (i + 1); j++) {
                hidari += " ";
                migi += " ";
            }
            String moji = "";
            int mojisuu = 1 + (2 * i);
            for (int j = 0; j < mojisuu; j++) {
                moji += "*";
            }
            System.out.println(hidari + moji + migi);
        }
    }
}

ふぅ〜、なんとかできた。こんなもんだよね？他の人はできた

もし新人女子 Java プログラマが『Groovy イン・アクション』を読んだら 〜第１章〜

のかなぁ〜、ちょっと遊びに行ってみようかな。もかはこなつの
席に近寄ってみた。すると苦悶の表情を浮かべているこなつがそ
こにいた。

　もか「こ...こなつ？大丈夫？苦しそうだけど...」
こなつ「え、あ、はい...ちょっと考え過ぎちゃって頭がぐるぐ

るなんです...」

　もか「そっかそっか...まだ時間はゆっくりあるし、頑張って
考えてみるといいよ」

こなつ「はい...もかちゃんはもう出来たの？」
　もか	「うん、まぁね。あれであってるかどうかは分からないけ

ど」

こなつ「やっぱりもかちゃんすごいね！他の誰よりも早いも
の...すごいです！」

こなつが突然大きな声を出すので、みんなの視線が二人の方に
集まった。

　もか「（小声で）こ、コラー（笑）恥ずかしいじゃん」
こなつ「（小声で）ごめんなさい（笑）」

途中でこなつのコードをちらっと見せてもらったのだが、こな
つは各段を全部打ち込んでしまっているようで ...

String hoshi4_1 = "   *   ";
String hoshi4_2 = "  ***  ";
String hoshi4_3 = " ***** ";
String hoshi4_4 = "*******";
String hoshi5_1 = "    *    ";
String hoshi5_2 = "   ***   ";
String hoshi5_3 = "  *****  ";
String hoshi5_4 = " ******* ";
String hoshi5_5 = "*********";
if (args[0].equals("4")) {
    System.out.println(hoshi4_1);
    System.out.println(hoshi4_2);
    System.out.println(hoshi4_3);
    System.out.println(hoshi4_4);
} else if (args[0].equals("5")) {
            .
            .
            .
}

のような感じになってしまっていた。これだともっといろんな
段数のパターン全部作るわけにはいかないよね？と教えてあげた

ら数十秒固まったあとになんとか納得してくれたようで、こなつ
はまた苦渋の表情を浮かべながらソースコードとにらめっこを始
めたのだった。

■もし Groovy だったら
自分の席に戻ったもかは課題をちょっとだけ早く終わったこと

で空いた時間、昨日のことを思い出していた。そういえばこの問
題、もしかしたら Groovy で書いたらもっと簡単に書けるんじゃ
ないのかな？ええと確か、そうそう、center っていうのがあった
わよね。たとえば、1 段目なんて、"*".center(7) とかすればでき
るんじゃないかしら？わくわくしてきて、もかはそのあとずっと
Groovy のことを考えていた。

その日の研修が終わり定時になると、もかは誰とも話をせずに
書店に走っていた。目的はもちろん一つ、『Groovy イン・アクショ
ン』を手に入れることだった。書店に到着したもかはさっそく、
昨日『Groovy イン・アクション』を見つけた棚へと走った。

「はぁ...はぁ...、あ、あったわ！」

そんなに焦らなくてもよかったのかもしれないけれど、もかは
本がそこにあったことで安堵した。あらためて本を手に取るもか、
踊り子さんも微笑んでくれている。よーし買うぞぉ！もかは値段
を確認してみた。本の裏表紙には ¥5,800 + 税と書かれていた。

「うわっ...ちょっと高い...けど！これはきっと運命なんだわ！」

興奮していたので、ちょっとだけ声が漏れてしまったようだ。
周りの人に白い目で見られたが、もかはそんなことを気にはして
いなかった。

もし新人女子 Java プログラマが『Groovy イン・アクション』を読んだら 〜第１章〜

■自室にて
早 速 買 っ て 来 た『Groovy イ ン・ ア ク シ ョ ン 』 を 片 手 に ま

ず は Groovy の イ ン ス ト ー ル 作 業 を 行 っ た。 こ れ は P536 を 見
ながらもかにも簡単にできた。その後、P15 に書かれていた
groovyConsole というもので、簡単にコード片を動かして試せる
ことが分かったので、もかは今日の講義中に考えていた Groovy
のコードを早速書いて実行させてみた。

int dan = 4

for (int i = 0; i < dan; i++) {

    String moji = "*" * (1 + i * 2)

    println(moji.center(dan * 2 - 1))

}

出来上がったコードはこれ。動かしてみたもかは感動した。

「わぁ〜！ちゃんとピラミッドが出て来た！やった〜！」

それにしてもすごいわ ... たった 5 行で出来ちゃった。それに、
すごく見やすいし、分かりやすい！

こうして、『Groovy イン・アクション』ともかは出会ったので
あった。

その日、遅くまでもかの部屋の明かりは消えることが無かった。

（続く）

もし新人女子 Java プログラマが『Groovy イン・アクション』を読んだら 〜第１章〜

はじめに
今回紹介する Grails プラグインは、MongoDB GORM プラグイ

ンです。
このプラグインをインストールすることで、GORM API を使っ

て Grails から MongoDB にアクセスできるようになります。
今回は、以下のバージョンの Grails、MongoDB で試しました。
•	 Grails : 1.3.7
•	 MongoDB : 1.8.0
•	 OS : MacOS X 10.6

MongoDBとは
MongoDB は、C++ で開発されたドキュメント指向のデータ

ベースです。BSON とよばれる JSON ライクなバイナリドキュメ
ントを管理します。

•	 詳細は、MongoDB の公式サイト http://www.mongodb.org/
を参照してください。

MongoDB GORMプラグインのインストール
最初に、Grails アプリケーションのプロジェクトに MongoDB 

GORM プラグインをインストールします。

$grails install-plugin mongodb

設定
何の設定もしない場合、MongoDB が 27017 ポートで動作して

いるとみなし、Grails アプリケーションは動作します。
MongoDB の 設 定 を し た い 場 合、grails-app/config/Config.

groovy を編集します。

grails {

    mongo {

        host = "localhost"

        port = 27107

        username = "jggug"

        password = "jggugpass"

        databaseName = "jggugdb"

    }

}

DB 名は、指定しない場合、Grails アプリケーション名となりま
す。

RDBMSとMongoDB を両方使いたい場合
Hibernate プ ラ グ イ ン を ア ン イ ン ス ト ー ル せ ず、MongoDB 

GORM プラグインと両方のプラグインがインストールされている
場合、grails-app/domain ディレクトリ配下のドメインクラスは、
Hibernate 経由で RDBMS に永続化されます。

特定のドメインクラスを MongoDB に永続化したい場合、ドメ
インクラスに mapWith プロパティで指定する必要があります。

static mapWith = "mongo"

あるいは、Hibernate のエンティティに追加された "mongo" と
いうスコープを使用します。

// Hibernate経由でRDBMSからインスタンスを取得
def hibernateBook = Book.get(1)

// RDBMSからインスタンスをMongoDBに保存
hibernateBook.mongo.save()

// MongoDBからインスタンスを取得
def mongoBook = Book.mongo.get(1)

Grails Plugin 探訪
第 3 回

~ MongoDB GORM プラグイン ~ 
URL: http://grails.org/plugin/mongodb

プラグインのバージョン : 1.0-M5
対応する Grails のバージョン : 1.3.5 以上

杉浦孝博

最近は Grails を使用したシステムの保守をしている自称プログラマ。
日本 Grails/Groovy ユーザーグループ事務局長。
共著『Grails 徹底入門』、共訳『Groovy イン・アクション』

Grails Plugin 探訪　第 3 回〜 MongoDB GORM プラグイン〜

http://www.mongodb.org/
http://grails.org/plugin/mongodb

MongoDB だけ使う場合やメインの DBとし
て使う場合

MongoDB だけ使う場合やメインの DB として使う場合、デフォ
ルトでインストールされている Hibernate プラグインをアンイン
ストールします。

$ grails uninstall-plugin hibernate

こうすることで、grails-app/domain ディレクトリ配下のドメ
インクラスは、MongoDB に永続化されるようになります。

ただし、Hibernate プラグインをアンインストールし GORM 
API を使う場合、NullPointerException が発生することがあるか
もしれません。

そのような場合は、ドメインクラスに mapWith プロパティを
設定し明示的に MongoDB を使うことを宣言してください。

高度な設定
設定は Config.groovy で行いますが、先程紹介した以外にも設

定できる項目があります。
options ブロックで、MongoDB との接続設定を変更できます。

grails {

    mongo {

        options {

            autoConnectRetry = true

            connectTimeout = 300

        }

    }

}

また、本番環境ではマスター / スレーブ、レプリケーションと
いった形式、複数の MongoDB サーバーを動作させることがある
と思います。

そのような設定も、Config.groovy で行います。

grails {

    mongo {

        replicaSet = [ "localhost:27017", 

"localhost:27018"]

    }

}

簡単なサンプル
ここでは、MongoDB を使った簡単なサンプルを作ってみます。

■Grails アプリケーションの作成
Grails ア プ リ ケ ー シ ョ ン を 作 成 し ま す。 こ こ で は、

pluginmongodb という名前とします。

$ grails create-app pluginmongodb

$ cd pluginmongodb

■ドメインクラスの作成
ドメインクラスを作成します。本をあらわす Book ドメインク

ラスと、出版社をあらわす Publisher ドメインクラスを作成しま
す。

$ grails create-domain-class Book

$ grails create-domain-class Publisher

Book ドメインクラスの内容は次のとおりです。

package pluginmongodb

class Book {

    String title

    Integer price

    Publisher publisher

    static constraints = {

    }

    static mapWith = "mongo"

    String toString() {

        title

    }

}

Publisher ドメインクラスの内容は次のとおりです。

package pluginmongodb

class Publisher {

    String name

    static constraints = {

    }

    static mapWith = "mongo"

    String toString() {

        name

    }

}

■コントローラとビューのスカフォールド
generate-all コマンドで、コントローラとビューをスカフォー

ルドします。

$ grails generate-all pluginmongodb.Book

$ grails generate-all pluginmongodb.Publisher

スカフォールドされるコントローラおよびビューは、GORM 
API(list, count, save など ) が使われた通常のファイルとなります。

G*Magazine vol.2

Grails Plugin 探訪　第 3 回〜 MongoDB GORM プラグイン〜



G*Magazine vol.2

23

■アプリケーションの実行
ア プ リ ケ ー シ ョ ン を 実 行 し ま す。 ア プ リ ケ ー シ ョ ン は、

MongoDB に対しドメインクラスのインスタンスの CRUD を行い
ます。

[ 一覧画面 ] 

[ 詳細画面 ] 

低水準な API
MongoDB GORM プラグインは、GORM API だけでなく、低水

準な API も提供しています。
Gmongo と呼ばれる Mongo Java Driver の Groovy ラッパーが

あり、コレクションに簡単にアクセスできるように拡張されてい
ます。

「mongo」という名前で Bean が定義されていますので、コン
トローラやサービスにインジェクションできます。

class BookService {

    def mongo

    ....

}

getDB() メソッドで DB の参照を取得し、その後コレクションに

対して操作を行います。

class BookService {

    def mongo

    def addBook(params) {

        def db = mongo.getDB("jggugdb")

        db.books.insert([title:params.title, 

price:params.price, publisher:params.publisher])

    }

}

Config.groovy の grails.mongo.databaseName で定義した DB 名
に "DB" を追加した名前の、DB を参照する Bean も定義されます。

また、insert() メソッドの代わりに、左シフト演算子が使用で
きます。

上記の例は、以下のようになります。

class BookService {

    def jggugdbDB

    def addBook(params) {

        jggugdbDB.books << [title:params.title, 

price:params.price, publisher:params.publisher]

    }

}

トランザクションについて
MongoDB は直接トランザクションをサポートしていません

が、MongoDB GORM プラグインを使用することで、トランザク
ション処理を行うことができます。

トランザクションの指定は、次のとおりです。
1.	サービスクラスにトランザクション指定をする
2.	withTransaction() メソッドを使う

■サービスクラスにトランザクション指定をする
サービスクラスの transactional プロパティに 'mongo' という値

を設定することで、トランザクション処理が行われます。

class BookService {

    static transactional = 'mongo'

    ...

}

■withTransaction() メソッドを使う
ドメインクラスの withTransaction() メソッドを使うことでも、

トランザクション処理が行われます。部分的なトランザクション
処理やコントローラでのトランザクション処理などに使用できま
す。

Book.withTransaction { status ->

    new Book(title:'Groovy in Action', 

price:5000).save()

    throw new RuntimeException("bad")

    new Book(title:'Grails in Action', 

price:5500).save()

}

おわりに
いろいろな NoSQL DB があり、その一つとして MongoDB が注

目されています。Grails アプリケーションから MongoDB にアク
セスする場合、MongoDB GORM プラグインを使ってみてはいか
がでしょうか。

リリース情報 2011.05.20
Grails
Grails は、Groovy や Hibernate などをベースとしたフルスタック
の Web アプリケーションフレームワークです。
URL: http://grails.org/
バージョン : 1.2.5, 1.3.7
■更新情報

•	 1.3.7 では、Groovy が 1.7.8 に更新されたり、スタックトレー
スログに出力されるリクエストパラメータの値をマスキング
したり、といった改良が行われています。

•	 1.3.7 リリースノート :  
http://www.grails.org/1.3.7+Release+Notes

Groovy
Groovy は、JavaVM 上で動作する動的言語です。
URL: http://groovy.codehaus.org/
バージョン : 1.7.10, 1.8.0
■更新情報

•	 1.7.10 では、Groovydoc の子クラスで package-templates、
doc-templates、class-templates がオーバーライド可能になっ
たり、いくつかのバグフィックスが行われています。

•	 1.7.10 リリースノート : http://jira.codehaus.org/secure/
ReleaseNote.jspa?projectId=10242&version=17229

•	 1.8.0 では、Groovy の動的な表現力の向上や、JSON をネイティ
ブサポート、GPars ライブラリのバンドルなどが行われてい
ます。

•	 1.8.0 リリースノート : http://docs.codehaus.org/display/
GROOVY/Groovy+1.8+release+notes

Griffon
Griffon は、デスクトップアプリケーションを開発するためのア
プリケーションフレームワークです。
URL: http://griffon.codehaus.org/
バージョン : 0.9.3-beata-1
■更新情報

•	 0.9.3-beata-1 では、新しい archetype が追加されたり、依存
する Groovy のバージョンが 1.8.0 に、Gant のバージョンが
1.9.5 にバージョンアップされたり、いくつか改良されてい
ます。

•	 0.9.3-beata-1 リリースノート : http://griffon.codehaus.org/
Griffon+0.9.3-beta-1

Gant
Gant は、XML の代わりに Groovy で Ant タスクを記述し実行する
ビルド管理ツールです。
URL: http://gant.codehaus.org/
バージョン : 1.9.5
■更新情報

•	 1.9.5 の変更箇所は不明です。
•	 1.9.5 リリースコメント : http://groovy.329449.n5.nabble.

com/Gant-1-9-5-released-2011-05-03T09-30-01-00-
td4366728.html

Gradle
Gradle は、Groovy でビルドスクリプトを記述し実行するビルド
管理ツールです。
URL: http://www.gradle.org/
バージョン : 0.9.2, 1.0-milestone-3

■更新情報
•	 1.0-milestone-3 では、Ivy リポジトリの定義が簡単になった

り、新しい API が追加されたり、いくつかのバグフィックス
が行われています。

•	 1.0-milestone-3 リリースノート : http://wiki.gradle.org/
display/GRADLE/Gradle+1.0-milestone-3+Release+Notes

Gaelyk
Gaelyk は、Groovy で記述する Google App Engine for Java 用のラ
イトウェイトなフレームワークです。
URL: http://gaelyk.appspot.com/
バージョン : 0.7
■更新情報

•	 0.7 では、Groovy が 1.8.0 に、GAE SDK が 1.5 にアップグレー
ドされたり、File サービスがサポートされるようになりまし
た。

•	 0.7 リリースノート : http://gaelyk.appspot.com/download

Google App Engine SDK for Java
Google App Engine SDK for Java は、Java で Google App Engine
用の Web アプリケーションを開発するための SDK です。
URL: http://code.google.com/intl/ja/appengine/
バージョン : 1.5.0
■更新情報

•	 1.5.0 では、Backend がサポートされたり、Task キューの pull
モードがサポートされたり、いくつかのバグフィックスが行
われています。

•	 1.5.0 リリースノート : http://code.google.com/p/
googleappengine/wiki/SdkForJavaReleaseNotes

GPars
GPars は、Groovy に直感的で安全な並行処理を提供するシステム
です。
URL: http://gpars.codehaus.org/
バージョン : 0.11GA、0.12-beta-1
■更新情報

•	 0.12-beta-1 では、構成可能な非同期機能や、Active Object
が追加されました。

Groovy++
Groovy++ は、Groovy 言語に対して静的な機能を拡張します。
URL: http://code.google.com/p/groovypptest/
バージョン : 0.4.243
■更新情報

•	 0.4.243 での変更点は不明です。
•	 0.4.243 リリースメッセージ : http://groups.google.com/

group/groovyplusplus/browse_thread/thread/52a7fb1dca4
3783d/632913b0785b02f8?show_docid=632913b0785b02f8

Spock
Spock は、Java や Groovy 用のテストと仕様のためのフレームワー
クです。
URL: http://code.google.com/p/spock/
バージョン : 0.5

GroovyServ
GroovyServ は、Groovy 処理系をサーバとして動作させることで
groovy コマンドの起動を見た目上高速化するものです。
URL: http://kobo.github.com/groovyserv/
バージョン : 0.7

■更新情報
•	 0.7 では、groovyserver の起動時に CLASSPATH 情報を表示し

たり (MacOS X, Linux のみ )、-server オプションを使用するよ
うにしたり、バグフィックスがいくつか行われています。

•	 0.7 チェンジログ : http://kobo.github.com/groovyserv/
changelog.html#0.7

Geb
Geb は、Groovy を使用した Web ブラウザを自動化する仕組みで
す。
URL: http://geb.codehaus.org/
バージョン : 0.5.1

Easyb
Easyb は、 ビ ヘ イ ビ ア 駆 動 開 発 (Behavior Driven 
Development:BDD) 用のフレームワークです。
URL: http://www.easyb.org/
バージョン : 0.9.8

Gmock
Gmock は、Groovy 用のモック・フレームワークです。 
URL: http://code.google.com/p/gmock/
バージョン : 0.8.1

HTTPBuilder
HTTPBuilder は、HTTP ベースのリソースに簡単にアクセスするた
めの方法です。
URL: http://groovy.codehaus.org/modules/http-builder/
バージョン : 0.5.1

CodeNarc
CodeNarc は、Groovy 向けの静的コード解析ツールです。
URL: http://codenarc.sourceforge.net/
バージョン : 0.13

GMetrics
GMetrics は、Groovy ソースコードのサイズや複雑さを計算した
り報告するためのツールです。
URL: http://gmetrics.sourceforge.net/
バージョン : 0.3

GContracts
GContracts は、Groovy で契約プログラミングを行うためのフレー
ムワークです。
URL: https://github.com/andresteingress/gcontracts
バージョン : 1.2.3

▼ぐるーびーたん　第１話のあらすじ
2011 年春、熊本からプログラマを目指して上京し

てきたぐるーびーたん。就職先のソフトウェア開発会
社（㈱ジーアスターソフト）の扉を勢いよく開けて初
出社！しかし、そこには寝袋にくるまった怪しい人影
が。。。
※この作品は、たいがいフィクションです。実在の人物、

団体とは関係ありません。

